

# INDEX function

[+ Show All](#)

This article describes the formula syntax and usage of the **INDEX function** in Microsoft Excel.

## Description

---

Returns a value or the reference to a value from within a table or range. There are two forms of the INDEX function: the **array** form and the reference form.

### If you want to

Return the value of a specified cell or array of cells

Return a reference to specified cells

### Then see

[Array form](#)

[Reference form](#)

## Array form

---

### DESCRIPTION

Returns the value of an element in a table or an **array**, selected by the row and column number indexes.

Use the array form if the first argument to INDEX is an array constant.

### SYNTAX

```
INDEX(array, row_num, [column_num])
```

The INDEX function syntax has the following **arguments**:

- **Array** Required. A range of cells or an array constant.
  - If array contains only one row or column, the corresponding row\_num or column\_num argument is optional.
  - If array has more than one row and more than one column, and only row\_num or column\_num is used, INDEX returns an array of the entire row or column in array.
- **Row\_num** Required. Selects the row in array from which to return a value. If row\_num is omitted, column\_num is required.
- **Column\_num** Optional. Selects the column in array from which to return a value. If column\_num is omitted, row\_num is required.

## REMARK

- If both the row\_num and column\_num arguments are used, INDEX returns the value in the cell at the intersection of row\_num and column\_num.
- If you set row\_num or column\_num to 0 (zero), INDEX returns the array of values for the entire column or row, respectively. To use values returned as an array, enter the INDEX function as an **array formula** in a horizontal range of cells for a row, and in a vertical range of cells for a column. To enter an array formula, press CTRL+SHIFT+ENTER.
- Row\_num and column\_num must point to a cell within array; otherwise, INDEX returns the #REF! error value.

## EXAMPLES

**Example 1**

The example may be easier to understand if you copy it to a blank worksheet.

[+](#) How do I copy an example?

	<b>A</b>	<b>B</b>
<b>1</b>	<b>Data</b>	<b>Data</b>
<b>2</b>	Apples	Lemons
<b>3</b>	Bananas	Pears
<b>4</b>	<b>Formula</b>	<b>Description (Result)</b>
<b>5</b>	=INDEX (A2:B3,2,2)	Value at the intersection of the second row and second column in the range (Pears)
<b>6</b>	=INDEX (A2:B3,2,1)	Value at the intersection of the second row and first column in the range (Bananas)

**Example 2**

The example may be easier to understand if you copy it to a blank worksheet.

[+](#) How do I copy an example?

	<b>A</b>	<b>B</b>
<b>1</b>	<b>Formula</b>	<b>Description (Result)</b>
	=INDEX({1,2;3,4},0,2)	Value in the first row, second column in the array constant (2)
<b>2</b>		Value in the second row, second column in the array constant (4)
<b>3</b>		

**NOTE** The formula in the example must be entered as an array formula. After copying the example to a blank worksheet, select the range A2:A3 starting with the formula cell. Press F2, and then press CTRL+SHIFT+ENTER. If the formula is not entered as an array formula, the single result is 2.

[↑ TOP OF PAGE](#)

## Reference form

---

### DESCRIPTION

Returns the reference of the cell at the intersection of a particular row and column. If the reference is made up of nonadjacent selections, you can pick the selection to look in.

### SYNTAX

```
INDEX(reference, row_num, [column_num], [area_num])
```

The INDEX function syntax has the following **arguments**:

- **Reference** Required. A reference to one or more cell ranges.
  - If you are entering a nonadjacent range for the reference, enclose reference in parentheses.
  - If each area in reference contains only one row or column, the row\_num or column\_num argument, respectively, is optional. For example, for a single row reference, use INDEX(reference,,column\_num).
- **Row\_num** Required. The number of the row in reference from which to return a reference.
- **Column\_num** Optional. The number of the column in reference from which to return a reference.
- **Area\_num** Optional. Selects a range in reference from which to return the intersection of row\_num and column\_num. The first area selected or entered is numbered 1, the second is 2, and so on. If area\_num is omitted, INDEX uses area 1.
  - For example, if reference describes the cells (A1:B4,D1:E4,G1:H4), then area\_num 1 is the range A1:B4, area\_num 2 is the range D1:E4, and area\_num 3 is the range G1:H4.

### REMARK

- After reference and area\_num have selected a particular range, row\_num and column\_num select a particular cell: row\_num 1 is the first row in the range, column\_num 1 is the first column, and so on. The reference returned by INDEX is the intersection of row\_num and column\_num.
- If you set row\_num or column\_num to 0 (zero), INDEX returns the reference for the entire column or row, respectively.

- Row\_num, column\_num, and area\_num must point to a cell within reference; otherwise, INDEX returns the #REF! error value. If row\_num and column\_num are omitted, INDEX returns the area in reference specified by area\_num.
- The result of the INDEX function is a reference and is interpreted as such by other formulas. Depending on the formula, the return value of INDEX may be used as a reference or as a value. For example, the formula CELL("width",INDEX(A1:B2,1,2)) is equivalent to CELL("width",B1). The CELL function uses the return value of INDEX as a cell reference. On the other hand, a formula such as 2\*INDEX(A1:B2,1,2) translates the return value of INDEX into the number in cell B1.

## EXAMPLE

The example may be easier to understand if you copy it to a blank worksheet.

[+](#) How do I copy an example?

	A	B	C
1	<b>Fruit</b>	<b>Price</b>	<b>Count</b>
2	Apples	0.69	40
3	Bananas	0.34	38
4	Lemons	0.55	15
5	Oranges	0.25	25
6	Pears	0.59	40
7			
8	Almonds	2.80	10
9	Cashews	3.55	16
10	Peanuts	1.25	20
11	Walnuts	1.75	12
12	<b>Formula</b>	<b>Description (Result)</b>	
	=INDEX(A2:C6, 2, 3)	The intersection of the second row and third column in the range A2:C6, which is the content of cell C3. (38)	
13	=INDEX((A1:C6, A8:C11), 2, 2, 2)	The intersection of the second row and second column in the second area of A8:C11, which is the content of cell B9. (3.55)	
14	=SUM(INDEX(A1:C11, 0, 3, 1))	The sum of the third column in the first area of the range A1:C11, which is the sum of C1:C6. (216)	
15	=SUM(B2:INDEX(A2:C6, 5, 2))	The sum of the range starting at B2, and ending at the intersection of the fifth row and the second column of the range A2:A6, which is the sum of B2:B6. (2.42)	
16			

 [TOP OF PAGE](#)

**See Also**

- [Lookup and reference functions \(reference\)](#)